
> restart:

> with(OreModules):

> with(OreMorphisms);

> with(linalg):

Let us consider another model of a tank containing a fluid and subjected to a one-dimensional horizontal
move studied in N. Petit, P. Rouchon, “Dynamics and solutions to some control problems for water-tank
systems”, IEEE Trans. Automatic Control, 47 (2002), 595-609. The system matrix is defined by:

> A:=DefineOreAlgebra(diff=[d,t],dual_shift=[delta,s],polynom=[t,s],
> comm=[alpha]):

> R:=matrix(2,3,[d,-d*delta^2,alpha*d^2*delta,d*delta^2,-d,alpha*d^2*delta]);

R :=

[
d −d δ2 α d2δ

d δ2 −d α d2δ

]
Let us consider the A = Q(α)[d, δ]-module M = A1×3/(A1×2 R) finitely presented by the matrix R and
let us compute the A-module structure of the endomorphism ring E = endA(M) of M :

> Endo:=MorphismsConstCoeff(R,R,A):

The A-module E is finitely generated by the endomorphisms fi’s defined by fi(π(λ)) = π(λ Pi), where
π : A1×3 −→ M denotes the projection onto M , λ ∈ A1×3 and Pi is one of the following matrices:

> Endo[1];

[


0 1 0

1 0 0

0 0 −1

 ,


0 0 0

0 0 0

δ2 −1 α d δ

 ,


1 0 0

0 1 0

0 0 1

 ,


0 0 0

0 0 0

1− δ2 1− δ2 0

 ,


0 0 0

−1 + δ2 −1 + δ2 0

0 0 0

 ,


0 0 0

α d α d 0

δ δ 0

 ,


0 0 α d δ

1 −δ2 0

0 0 −δ2 − 1

 ,


0 0 0

1 −δ2 α d δ

0 0 0

]

The generators fi’s of E satisfy the following A-linear relations

> Endo[2]; 

−d 0 d δ2 0 0 0 d 0

d δ2 0 −d 0 0 0 −d 0

0 d 0 0 0 0 0 0

0 0 0 d 0 0 0 0

0 0 0 δ 0 −1 + δ2 0 0

0 0 0 0 d 0 0 0

0 0 0 0 0 0 0 d


i.e., if we denote by F = (f1 . . . f8)T , we then have Endo[2]F = 0.

The multiplication table Endo[3] of the generators fi’s gives us a way to rewrite the composition fi ◦fj in
terms of A-linear combinations of the fk’s or, in other words, if we denote by ⊗ the Kronecker product,
namely, F ⊗F = ((f1◦F )T . . . (f8◦F )T )T , then the multiplication table T of the generators fj ’s satisfies
F ⊗ F = T F , where T is the matrix Endo[3] without the first column which corresponds to the indices
(i, j) of the product fi ◦ fj . We do not print here this matrix as it belongs to A64×8. We can use it
for rewriting any polynomial in the fi’s with coefficients in A in terms of a A-linear combination of the
generators fj ’s.

Let us now try to compute idempotents of E defined by idempotent matrices, namely, elements e ∈ E
satisfying e2 = e and defined by matrices P ∈ A3×3 and Q ∈ A2×2 satisfying the relations R P = QR,
P 2 = P and Q2 = Q:

> Idem:=IdempotentsMatConstCoeff(R,Endo[1],A,0);

Idem := [[


1/2 1/2 0

1/2 1/2 0

−c51
(
−1 + δ2

)
−c51

(
−1 + δ2

)
0

 ,


0 0 0

0 0 0

0 0 0

 ,


1 0 0

0 1 0

0 0 1

 ,


0 0 0

−δ2 1 −α δ d

0 0 0

 ,


1 0 0

δ2 0 α δ d

0 0 1

 ,


1/2 −1/2 0

−1/2 1/2 0

−c51
(
−1 + δ2

)
−c51

(
−1 + δ2

)
1

],

[Ore algebra, [“diff ′′, dual shift ], [t, s], [d, δ], [t, s], [α, c51 ], 0, [], [], [t, s], [], [], [diff = [d, t],

dual shift = [δ, s]]]]

Let us consider the first entry P1 of Idem[1] where we have set the arbitrary constant c51 to 0 and the
matrix Q1 ∈ A2×2 satisfying R P1 = Q1 R:

> P[1]:=subs(c51=0,evalm(Idem[1,1])); Q[1]:=Factorize(Mult(R,P[1],A),R,A);

P1 :=


1/2 1/2 0

1/2 1/2 0

0 0 0

 Q1 :=

[
1/2 −1/2

−1/2 1/2

]

As the entries of the matrices P1 and Q1 belong to Q, using linear algebraic techniques, we can easily
compute bases of the free A-modules kerA(.P1), kerA(.Q1), imA(.P1) = kerA(.(I3 −P1)) and imA(.Q1) =
kerA(.(I2 −Q1)):

> U1:=SyzygyModule(P[1],A): U2:=SyzygyModule(evalm(1-P[1]),A):
> U:=stackmatrix(U1,U2);
> V1:=SyzygyModule(Q[1],A): V2:=SyzygyModule(evalm(1-Q[1]),A):
> V:=stackmatrix(V1,V2);

U :=


1 −1 0

0 0 1

1 1 0

 V :=

[
1 1

1 −1

]

We can check that J1 = U P1 U−1 and J2 = V Q1 V −1 are block-diagonal matrices formed by the matrices
0n and Im:

> VERIF1:=Mult(U,P,LeftInverse(U,A),A);
> VERIF2:=Mult(V,Q,LeftInverse(V,A),A);

VERIF1 :=


0 0 0

0 0 0

0 0 1

 VERIF2 :=

[
0 0

0 1

]

Then, the matrix R is equivalent to the following block-diagonal matrix V R U−1:

> R_dec:=map(factor,simplify(Mult(V,R,LeftInverse(U,A),A)));

R dec :=

[
d

(
δ2 + 1

)
2 α d2δ 0

0 0 −d (δ − 1) (δ + 1)

]

This last result can be directly obtained by means of the command HeuristicDecomposition:

> map(factor,HeuristicDecomposition(R,P[1],A)[1]);[
d

(
δ2 + 1

)
2 α d2 δ 0

0 0 −d (δ − 1) (δ + 1)

]
We can use another idempotent matrix P2 listed in Idem[1] to obtain another decomposition of the
matrix R. Let us consider the fourth one and the corresponding idempotent matrix Q2:

> P[2]:=Idem[1,4]; Q[2]:=Factorize(Mult(R,P[2],A),R,A);

P2 :=


0 0 0

−δ2 1 −α δ d

0 0 0

 Q2 :=

[
0 δ2

0 1

]

As we have P 2
2 = P2 and Q2

2 = Q2, we know that the A-modules kerA(.P2), kerA(.Q2), imA(.P2) =
kerA(.(I3 − P2)) and imA(.Q2) = kerA(.(I2 − Q2)) are projective, and thus, free by the Quillen-Suslin
theorem. Let us compute basis of those free A-modules:

> U11:=SyzygyModule(P[2],A): U21:=SyzygyModule(evalm(1-P[2]),A):
> UU:=stackmatrix(U11,U21);
> V11:=SyzygyModule(Q[2],A): V21:=SyzygyModule(evalm(1-Q[2]),A):
> VV:=stackmatrix(V11,V21);

UU :=


1 0 0

0 0 1

δ2 −1 α δ d

 VV :=

[
−1 δ2

0 1

]

As previously, we can check that the idempotent matrices P2 and Q2 are equivalent to block-diagonal
matrices formed by the matrices 0n and Im:

> VERIF1:=Mult(UU,P[1],LeftInverse(UU,A),A);
> VERIF2:=Mult(VV,Q[1],LeftInverse(VV,A),A);

VERIF1 :=


0 0 0

0 0 0

0 0 1

 VERIF2 :=

[
0 0

0 1

]

Then, the matrix R is equivalent to the following block-diagonal matrix:

> R_dec1:=map(factor,simplify(Mult(VV,R,LeftInverse(UU,A),A)));

R dec1 :=

[
d (δ − 1) (δ + 1)

(
δ2 + 1

)
α d2 δ (δ − 1) (δ + 1) 0

0 0 d

]
We can check this last result by means of the command HeuristicDecomposition:

> map(factor,HeuristicDecomposition(R,P[2],A)[1]);[
d (δ − 1) (δ + 1)

(
δ2 + 1

)
α d2 δ (δ − 1) (δ + 1) 0

0 0 d

]
Hence, we obtain another decomposition of the matrix R. If we denote by

T1 = (d (δ2 + 1) 2 α d2 δ),
T2 = d (δ2 − 1),
T3 = (d (δ2 − 1) (δ2 + 1) α d2 δ (δ2 − 1)),
T4 = d,


M1 = A1×2/(A T1),
M2 = A/(A T2),
M3 = A1×2/(A T3),
M4 = A/(A T4),

(1)

then we have the following decompositions of the A-module M :

M ∼= M1 ⊕M2, M ∼= M3 ⊕M4. (2)

Let us now study the A-module structure of E defined by A1×8/(A1×7 Endo[2]):

> ext1:=Exti(Involution(Endo[2],A),A,1): ext1[1];

d δ2 − d 0 0 0 0 0 0

0 d 0 0 0 0 0

0 0 d δ2 − d 0 0 0 0

0 0 0 d 0 0 0

0 0 0 0 d 0 0

0 0 0 0 0 d δ2 − d 0

0 0 0 0 0 0 d


> ext1[2]; 

1 0 1 0 0 0 0 0

0 1 0 0 0 0 0 0

0 0 δ2 + 1 0 0 0 1 0

0 0 0 1 0 0 0 0

0 0 0 0 1 0 0 0

0 0 0 0 0 1 0 0

0 0 0 0 0 0 0 1


Hence, the following torsion elements of E

t1 = f1 + f3,

t2 = f2,

t3 = (δ2 + 1) f3 + f7,

t4 = f4,

t5 = f5,

t6 = f6,

t7 = f8,



d (δ2 − 1) t1 = 0,

d t2 = 0,

d (δ2 − 1) t3 = 0,

d t4 = 0,

d t5 = 0,

d (δ2 − 1) t6 = 0,

d t7 = 0,

(3)

generate the A-module t(E) and we have E/t(E) = A1×8/(A1×7 ext1[2]). As the A-module E/t(E) is
torsion-free, it can be parametrized by means of the matrix ext1[3] defined by

> ext1[3]; 

1

0

−1

0

0

0

δ2 + 1

0

i.e., we have E/t(E) ∼= A1×8 ext1[3]. As ext1[3] admits a left-inverse over A defined by

> LeftInverse(ext1[3],A); [
0 0 −1 0 0 0 0 0

]
we obtain that A1×8 ext1[3] = A, i.e., E/t(E) is a free A-module of rank 1. Using that the short exact
sequence of A-modules 0 −→ t(E) ι−→ E

ρ−→ E/t(E) −→ 0 ends with a projective A-module, it splits
and we get E ∼= t(E)⊕ E/t(E) ∼= t(E)⊕A. Let us now study t(E).

> L:=Factorize(Endo[2],ext1[2],A);

L :=



−d 0 d 0 0 0 0

d δ2 0 −d 0 0 0 0

0 d 0 0 0 0 0

0 0 0 d 0 0 0

0 0 0 δ 0 −1 + δ2 0

0 0 0 0 d 0 0

0 0 0 0 0 0 d


> SyzygyModule(ext1[2],A);

INJ (7)

Lemma 3.1 of T. Cluzeau, A. Quadrat, “Factoring and decomposing a class of linear functional systems”,
Linear Algebra and Its Applications, 428 (2008), 324-381, we obtain that t(E) ∼= A1×7/(A1×7 L). From
the structure of the full row rank matrix L, we obtain that

t(E) ∼= [A/(A d)]3 ⊕A1×2/(A1×2 S1)⊕A1×2/(A1×2 S2),

where where N l denotes l direct sums of N and the matrices S1 and S2 are defined by:

> S[1]:=submatrix(L,1..2,[1,3]);

S1 :=

[
−d d

d δ2 −d

]
> S[2]:=submatrix(L,4..5,[4,6]);

S2 :=

[
d 0

δ −1 + δ2

]
Let us check whether or not the matrix S1 is equivalent to a block-diagonal matrix:

> E[1]:=MorphismsConstCoeff(S[1],S[1],A):

> Idem[1]:=IdempotentsMatConstCoeff(S[1],E[1][1],A,0,alpha);

Idem1 := [[

[
c31 −c31 + 1

c31 −c31 + 1

]
,

[
c31 −c31

c31 − 1 −c31 + 1

]
,

[
0 0

0 0

]
,

[
1 0

0 1

]
], [Ore algebra,

[“diff ′′, dual shift ], [t, s], [d, δ], [t, s], [α, c31 ], 0, [], [], [t, s], [], [], [diff = [d, t], dual shift = [δ, s]]]]

> X[1]:=subs(c31=0,evalm(Idem[1][1,1]));

X1 :=

[
0 1

0 1

]
We obtain that the matrix S1 is equivalent to the following block diagonal matrix:

> map(factor,HeuristicDecomposition(S[1],X[1],A)[1]);[
−d 0

0 d (δ − 1) (δ + 1)

]
Hence, we have A1×2/(A1×2 S1) ∼= A/(A d)⊕A/(A d (δ2 − 1)).

Let us check whether or not the matrix S2 is equivalent to a block-diagonal matrix:

> E[2]:=MorphismsConstCoeff(S[2],S[2],A):

> Idem[2]:=IdempotentsMatConstCoeff(S[2],E[2][1],A,0,alpha);

Idem2 := [[

[
0 0

0 0

]
,

[
1 0

0 1

]
], [Ore algebra, [“diff ′′, dual shift ],

[t, s], [d, δ], [t, s], [α], 0, [], [], [t, s], [], [], [diff = [d, t], dual shift = [δ, s]]]]

> X[2]:=Idem[2][1,1]; Y[2]:=diag(0$2); Z:=diag(0$2);

X2 :=

[
0 0

0 0

]
Y2 :=

[
0 0

0 0

]
Z :=

[
0 0

0 0

]
> Lambda:=RiccatiConstCoeff(S[2],X[2],Y[2],Z,A,1,alpha)[1];

Λ := [[

[
0 0

0 0

]
,

[
0 δ

0 −1

]
]]

> X_bar[2]:=simplify(evalm(X[2]+Mult(Lambda[2],S[2],A)));

X bar2 :=

[
δ2

(
−1 + δ2

)
δ

−δ 1− δ2

]
We obtain that the matrix S2 is equivalent to the following block-diagonal one:

> map(factor,HeuristicDecomposition(S[2],X_bar[2],A)[1]);[
d (δ − 1) (δ + 1) 0

0 1

]
In particular, we have A1×2/(A1×2 S2) ∼= A/(A d (δ2 − 1)), which shows that:

t(E) ∼= [A/(A d)]4 ⊕ [A/(A d (δ2 − 1))]2.

Hence, we obtain the following decomposition of the A-module E:

E ∼= [A/(A d)]4 ⊕ [A/(A d (δ2 − 1))]2 ⊕A. (4)

We now explicitly describe the previous isomorphism. Let us first compute a generalized inverse of the
matrix ext1[2] over A:

> W:=GeneralizedInverse(ext1[2],A);

W :=



1 0 0 0 0 0 0

0 1 0 0 0 0 0

0 0 0 0 0 0 0

0 0 0 1 0 0 0

0 0 0 0 1 0 0

0 0 0 0 0 1 0

0 0 1 0 0 0 0

0 0 0 0 0 0 1


We now introduce the matrix H = I6 −W ext1[2] defined by:

> H:=simplify(evalm(1-Mult(W,ext1[2],A)));

H :=



0 0 −1 0 0 0 0 0

0 0 0 0 0 0 0 0

0 0 1 0 0 0 0 0

0 0 0 0 0 0 0 0

0 0 0 0 0 0 0 0

0 0 0 0 0 0 0 0

0 0 −δ2 − 1 0 0 0 0 0

0 0 0 0 0 0 0 0


Using the fact that ext1[2]H = 0, we obtain that the A-morphism σ : E/t(E) −→ E defined by
σ(π′(λ)) = π(λ H), where π : A1×8 −→ E (resp., π′ : A1×8 −→ E/t(E)) denotes the canonical projection
onto E (resp., E/t(E)) and λ ∈ A1×8, satisfies ρ ◦ σ = idE/t(E). For more details, see Theorem 4 of
A. Quadrat, D. Robertz, “Parametrizing all solutions of uncontrollable multidimensional linear systems”,
Proceedings of 16th IFAC World Congress, Prague (Czech Republic), 04-08/07/05. If we denote by
{gi = ρ(fi)}i=1,...,8 a set of generators of the A-module E/t(E), then the A-morphism σ : E/t(E) −→ E
is defined by: 

σ(g1) = −f3,

σ(g2) = 0,

σ(g3) = f3,

σ(g4) = 0,

σ(g5) = 0,

σ(g6) = 0,

σ(g7) = −(δ2 + 1) f3,

σ(g8) = 0.

Using (3), the A-morphism χ : idE − σ ◦ ρ : E −→ E is then defined by:

χ(f1) = f1 + f3 = t1,

χ(f2) = f2 = t2,

χ(f3) = f3 − f3 = 0,

χ(f4) = f4 = t4,

χ(f5) = f5 = t5,

χ(f6) = f6 = t6,

χ(f7) = f7 + (δ2 + 1) f3 = t3,

χ(f8) = f8 = t7.

Hence, if we define the A-morphism κ : E −→ t(E) by

κ(f1) = t1,

κ(f2) = t2,

κ(f3) = 0,

κ(f4) = t4,

κ(f5) = t5,

κ(f6) = t6,

κ(f7) = t3,

κ(f8) = t7,

then we get the identity idE = σ ◦ ρ + ι ◦ κ. Therefore, we obtain:

f1 = t1 − idM ,

f2 = t2,

f3 = idM ,

f4 = t4,

f5 = t5,

f6 = t6,

f7 = t3 − (δ2 + 1) idM ,

f8 = t7.

We find that {t1, . . . , t7, idM} is the same set of generators of the A-module E as {fi}i=1,...,8. Hence, the
family of generators {t1, . . . , t7, idM} admits the same multiplication table Endo[3].

Let us show how to find again (4) from (2). Using the fact that M ∼= M1 ⊕M2, we get:

E = endA(M) ∼= endA(M1)⊕ homA(M1,M2)⊕ homA(M2,M1)⊕ endA(M2).

Using the fact that M2 = A/(A d (δ2 − 1)), we have endA(M2) = A/(A d (δ2 − 1)). With the notations
(1)

> T[1]:=submatrix(R_dec,1..1,1..2);

T1 :=
[

d
(
δ2 + 1

)
2 α d2δ

]
> T[2]:=submatrix(R_dec,2..2,3..3);

T2 :=
[
−d (δ − 1) (δ + 1)

]
we have homA(M1,M2) = A1×3/(A1×3 Morph[1][2]), where Morph[1][2] is defined by:

> Morph[1]:=MorphismsConstMorphCoeff(T[1],T[2],A): Morph[1][2];

−1 + δ2 −δ 0

0 d 0

0 0 d


Using the structure of the matrix Morph[1][2] and the previous decomposition of S2, we obtain:

homA(M1,M2) ∼= A/(d (δ2 − 1))⊕A/(A d).

Let us now compute homA(M2,M1):

> Morph[2]:=MorphismsConstCoeff(T[2],T[1],A);

Morph2 := [[
[

δ2 + 1 2 α d δ
]
],

[
d

]
]

We obtain that homA(M2,M1) is generated by one generator h satisfying the relation d h = 0, i.e., we
have homA(M2,M1) ∼= A/(A d).

We now need to characterize the A-module endA(M1):

> Morph[3]:=MorphismsConstCoeff(T[1],T[1],A): Morph[3][2];[
d 0 0

0 d δ2 + d d

]
Hence, we obtain endA(M1) ∼= A/(A d)⊕A1×2/(A J), where J ∈ A1×2 is defined by:

> J:=submatrix(Morph[3][2],2..2,2..3);

J :=
[

d δ2 + d d
]

Let us study the A-module N = A1×2/(A J):

> Extension1:= Exti(Involution(J,A),A,1);

Extension1 := [
[

d
]
,
[

δ2 + 1 1
]
,

[
−1

δ2 + 1

]
]

We get that t(N) = (A ((δ2 + 1) 1))/(A J) ∼= A/(A d) and N/t(N) = A1×2/(A ((δ2 + 1) 1)). The
A-module N/t(N) is free as its parametrization Extension1[3] admits a left-inverse over A:

> LeftInverse(Extension1[3],A); [
−1 0

]
Therefore, the short exact sequence 0 −→ t(N) −→ N −→ N/t(N) −→ 0 splits and we obtain that
N ∼= t(N)⊕N/t(N) ∼= A/(A d)⊕A, a fact proving that endA(M1) ∼= [A/(A d)]2 ⊕A and:

E ∼= endA(M1)⊕ homA(M1,M2)⊕ homA(M2,M1)⊕ endA(M2)
∼= [A/(A d)]2 ⊕A⊕A/(A d (δ2 − 1))⊕A/(A d)⊕A/(A d)⊕A/(A d (δ2 − 1))
∼= [A/(A d)]4 ⊕ [A/(A d (δ2 − 1))]2 ⊕A.

We can also use the second decomposition M ∼= M3 ⊕ M4 obtained in (2) to find again the previous
result. Indeed, we have:

E = endA(M) ∼= endA(M3)⊕ homA(M3,M4)⊕ homA(M4,M3)⊕ endA(M4).

Using similar techniques as the previous ones, we can prove that
endA(M3) ∼= [A/(A d (δ2 − 1))]2 ⊕A,

homA(M3,M4) ∼= [A/(A d)]2,
homA(M4,M3) ∼= A/(A d),
endA(M4) ∼= A/(A d),

which finally shows again that E ∼= [A/(A d)]4 ⊕ [A/(A d (δ2 − 1))]2 ⊕A.

